

Web-based Supporting Materials for “Bayesian Hierarchical  
Joint Modeling Using Skew-Normal/Independent  
Distributions” by Geng Chen and Sheng Luo

Table 1: Setting I: simulation results from models  $RM_N$ ,  $RM_T$  and  $RM_{ST}$  when there were skewness and 5% outliers in the continuous outcome.

	True	$RM_N$				$RM_T$				$RM_{ST}$			
		Bias	SD	SE	CP	Bias	SD	SE	CP	Bias	SD	SE	CP
$a_1$	25.000	<b>4.390</b>	0.774	0.713	<b>0.000</b>	<b>5.954</b>	0.618	0.584	<b>0.000</b>	0.421	1.055	1.074	0.941
$b_1$	10.000	<b>0.345</b>	0.801	0.657	<b>0.843</b>	<b>0.587</b>	0.461	0.458	<b>0.734</b>	-0.147	0.525	0.509	0.946
$\sigma_1$	5.000	<b>7.447</b>	1.332	0.376	<b>0.000</b>	<b>-1.227</b>	0.254	0.267	<b>0.013</b>	<b>-1.369</b>	0.247	0.257	<b>0.008</b>
$a_{21}$	-2.700	0.099	0.205	0.183	0.858	0.187	0.173	0.162	0.760	-0.045	0.197	0.187	0.925
$a_{22}$	-0.600	0.018	0.149	0.140	0.925	0.048	0.137	0.129	0.911	-0.029	0.148	0.142	0.936
$a_{23}$	2.000	-0.060	0.158	0.160	0.923	-0.101	0.153	0.147	0.883	-0.001	0.167	0.163	0.941
$a_{24}$	2.800	-0.069	0.187	0.183	0.915	-0.133	0.175	0.168	0.875	0.018	0.192	0.187	0.938
$a_{25}$	5.000	-0.074	0.345	0.316	0.918	-0.185	0.328	0.293	0.846	0.086	0.353	0.321	0.928
$a_{26}$	6.000	0.076	0.512	0.467	0.948	-0.056	0.487	0.445	0.932	0.256	0.518	0.473	0.922
$b_2$	2.000	-0.103	0.152	0.145	0.871	-0.211	0.124	0.119	0.594	-0.005	0.148	0.144	0.936
$a_{31}$	-0.100	0.009	0.074	0.076	0.954	0.015	0.073	0.074	0.940	0.004	0.074	0.075	0.949
$a_{32}$	1.000	0.012	0.082	0.083	0.954	0.016	0.081	0.082	0.945	0.010	0.083	0.082	0.952
$a_{33}$	1.800	0.012	0.096	0.103	0.961	0.016	0.096	0.102	0.958	0.009	0.096	0.102	0.971
$a_{34}$	2.600	0.032	0.142	0.140	0.951	0.036	0.141	0.140	0.953	0.031	0.142	0.140	0.968
$a_{35}$	3.300	0.069	0.196	0.193	0.938	0.074	0.198	0.193	0.940	0.068	0.197	0.193	0.941
$a_{36}$	4.000	0.149	0.285	0.278	0.928	0.154	0.291	0.279	0.924	0.149	0.289	0.279	0.928
$b_3$	0.400	0.004	0.060	0.060	0.954	-0.010	0.056	0.056	0.958	-0.007	0.057	0.056	0.946
$\beta_{10}$	0.400	<b>-1.459</b>	0.191	0.188	<b>0.000</b>	<b>-1.246</b>	0.163	0.160	<b>0.000</b>	<b>-1.278</b>	0.173	0.173	<b>0.000</b>
$\beta_{11}$	-0.500	<b>0.464</b>	0.214	0.207	<b>0.374</b>	<b>0.395</b>	0.182	0.184	<b>0.440</b>	<b>0.388</b>	0.194	0.199	<b>0.504</b>
$\rho$	0.400	-0.223	0.147	0.164	0.714	-0.217	0.105	0.109	0.438	-0.212	0.107	0.111	0.477
$\sigma_u$	1.300	-0.526	0.124	0.118	0.034	-0.417	0.090	0.090	0.018	-0.334	0.101	0.101	0.147
$\gamma$	-1.000	<b>0.304</b>	0.158	0.109	<b>0.273</b>	<b>0.303</b>	0.162	0.109	<b>0.286</b>	<b>0.305</b>	0.157	0.109	<b>0.273</b>
$\eta_1$	0.400	—	—	—	—	—	—	—	—	—	—	—	—
$\eta_2$	1.000	—	—	—	—	—	—	—	—	—	—	—	—

Note: Large bias and poor CR are highlighted in bold.

## JAGS code for fitting model $JM_{ST}$

```
model
{
  for (i in 1:obs) # obs: number of total observations
  {
    w1[i] ~ dgamma(nu,nu)
    Y.conti[i] ~ dnorm(mu.conti[i], tau[i]) # continuous outcome
    tau[i] <- w1[i]*tau.conti
    # two ordinal outcomes
    Y.ordi.1[i] ~ dcat(prob.y.1[i, 1:n.1])
    Y.ordi.2[i] ~ dcat(prob.y.2[i, 1:n.2])
  }

  # construct ST distribution on the mean of the continuous outcome
  for (i in 1:obs)
  {
    mu.conti[i] <- a.conti + b.conti * theta[i] + delta.sn*(w.sn[subject[i]])
  }

  # Construct the probability vector for the ordinal variables
  for (i in 1:obs)
  {
    for (l in 1:(n.1-1)) { logit(psi.1[i, l]) <- a.ordi.1[l] - b.ordi.1*theta[i] }
    prob.y.1[i, 1] <- psi.1[i, 1]
    for (l in 2:(n.1-1)) { prob.y.1[i, l] <- psi.1[i, l] - psi.1[i, l-1] }
    prob.y.1[i, n.1] <- 1 - psi.1[i, (n.1-1)]

    for (l in 1:(n.2-1)) { logit(psi.2[i, l]) <- a.ordi.2[l] - b.ordi.2*theta[i] }
    prob.y.2[i, 1] <- psi.2[i, 1]
    for (l in 2:(n.2-1)) { prob.y.2[i, l] <- psi.2[i, l] - psi.2[i, l-1] }
    prob.y.2[i, n.2] <- 1 - psi.2[i, (n.2-1)]
  }

  for (i in 1:N)
  {
    # construct random effects
    u[i, 1:2] ~ dnorm(zero[, ], precision[, ])
    # construct variable for the skewness parameter
    w.sn[i] ~ dnorm(0,1)I(0,)
  }

  # construct the variance-covariance matrix for random effects
  precision[1:2,1:2] <- inverse(sigma[, ])
  sigma[1,1] <- 1
  sigma[1,2] <- rho*sig1
  sigma[2,1] <- sigma[1,2]
  sigma[2,2] <- sig1*sig1

  # construct theta, the latent variable of subject i at time j
  for (i in 1:obs)
  {
    theta[i] <- u[subject[i], 1] + (beta[1] + beta[2]*treat[i]
      + u[subject[i], 2])*t[i]
  }

  # construct survival part
  for (i in 1:N)
  {
```

```

# use zero-trick to specify the likelihood
phi[i] <- -lL[i]
zeros[i] ~ dpois(phi[i])

# k is the number of time interval for baseline step function
for (k in 1:3) {
  h0[i,k] <- inprod(g[k],I0[i,k])
  gt[i,k] <- inprod(g[k],dt1[i,k])
}

## take log of the survival function
lh[i] <- gam*treat.pts[i] + omega2*u[i, 1] + omega3*u[i,2] + log(sum(h0[i,]))
lS[i] <- -(exp(gam*treat.pts[i] + omega2*u[i, 1] + omega3*u[i,2])*sum(gt[i,]))
# event=1 for event; 0 for censored
ll[i] <- event[i]*lh[i] + lS[i] -log(1.0E+08)
}

# prior for g
for (k in 1:3)
{
  g[k] ~ dunif(0,20)
}
# prior for parameters gam, omega2, and omega3
gam ~ dnorm(0, 0.01)
omega2 ~ dnorm(0, 0.01)
omega3 ~ dnorm(0, 0.01)

# prior for regression coefficients
for (i in 1:2)
{
  beta[i] ~ dnorm(0, 0.01)
}

# specify prior distributions
rho ~ dunif(-1, 1)
sig1 ~ dgamma(0.01, 0.01)

# prior for continuous variable's parameters
b.conti ~ dgamma(0.001,0.001)
a.conti ~ dnorm(0, 0.0005)
tau.conti ~ dgamma(0.001,0.001)
sd.conti <- 1/sqrt(tau.conti)
b.ordi.1 ~ dgamma(0.001,0.001)
b.ordi.2 ~ dgamma(0.001,0.001)

a.ordi.1[1] ~ dnorm(0,0.001)
for (l in 2:(n.1-1)) { a.ordi.1[l] <- a.ordi.1[l-1] + delta.1[l-1] }
for (i in 1:(n.1-2)) {delta.1[i] ~ dnorm(0,0.01)I(0,)}
a.ordi.2[1] ~ dnorm(0,0.001)
for (l in 2:(n.2-1)) { a.ordi.2[l] <- a.ordi.2[l-1] + delta.2[l-1] }
for (i in 1:(n.2-2)) {delta.2[i] ~ dnorm(0,0.01)I(0,)}

# prior distribution for df and skewness parameters
nu~dgamma(0.001,0.001)
delta.sn~dnorm(0,0.001)I(0,)}

```